
[bookmark: _Toc381225127]Построение информационной системы «Тест»

[bookmark: _Toc381225128]Цель работы

Используя Элементы управления и программы на VBA, создать интерфейс информационной системы (ИС) для тестирования студентов по теме лекций курса «История развития вычислительной техники». Для этого на листе Excel создать соответствующие кнопки и формы. База вопросов по теме курса с вариантами ответов заносится на один из листов. 
Разработанная программа должна выбирать случайным образом вопросы по выбранной теме, формировать «электронный билет» и принимать ответ от тестируемого на каждый вопрос билета. По специальным кнопкам можно проверить результат своего ответа или выйти из процесса тестирования.
В «билете» 3 вопроса. Один правильный ответ – оценка 3, два – 4, три – 5. Повторно пройти тестирование, даже по другой теме, нельзя.
Результаты тестирования записываются на отдельный лист.
Практическая работа рассчитана на 2 занятия, т.к. предполагает большой объем самостоятельного программирования.

[bookmark: _Toc381225129]I. Главная форма – создание кнопки для запуска тестирования

1. Открыть приложение Excel
2. Создать книгу с тремя листами (Рис. 3.1). Дайте название книге «ТЕСТ», каждому листу – «Главная форма», «Список студентов». 
3. В папке с файлом методической работы имеется файл «База_вопросов». Открыть этот файл. На вкладке листа «Вопросы» вызвать контекстное меню, перейти к команде «Переместить/Скопировать». В поле «Переместить выбранные листы в книгу» открывшегося диалогового окна выбрать имя созданной Вами книги, установить флажок «Создать копию» (!!! НЕ ЗАБУДЬТЕ про флажок, иначе вы УДАЛИТЕ исходный лист из чужой книги) и только тогда нажать кнопку «ОК».
4. На листе «Главная форма создать» кнопку «Тестирование» (рис. 3.1), используя на вкладке Разработчик в группе Элементы управления кнопку-команду Вставить  Элементы управления формы.

[image: ]

Рис. 3.1. Книга ИС «ТЕСТ»

5. Отредактировать надпись на кнопке, используя команду контекстного меню Формат объекта.
6. Используя команду Просмотр кода с ленточной вкладки Разработчик  Элементы управления создать процедуру обработки созданной кнопки на языке VBA:
Sub Кнопка1_Щелчок()
UserForm1.Show
End Sub
[bookmark: _GoBack]Кнопка открывает первую пользовательскую форму. Текст этой программы записывается в отдельный модуль редактора VBA: Insert ‑ Module.

[bookmark: _Toc381225130]II. Первая пользовательская форма – выбор темы для тестирования

1. Перейти в окно реактора VBA
Используя команду Insert горизонтального меню окна, создать форму UserForm (рис. 3.2). На неё встроить объекты Переключатель и Кнопка. 

Рис. 3.2. Проект ИС «ТЕСТ» и Форма1

Текст программ (с соответствующими комментариями), обрабатывающих встроенные объекты, следующий:

Private Sub CommandButton1_Click()
If OptionButton1.Value Then
  Part = 1
ElseIf OptionButton2.Value Then
  Part = 2
Else
MsgBox "Вы не выбрали тему!"
Exit Sub
End If
    Part_N ‘ программа инициализации ИС
End Sub

Private Sub CommandButton2_Click()
UserForm1.Hide
End Sub

РАЗОБРАТЬ И БЫТЬ ГОТОВЫМ ОБЪЯСНИТЬ ДЕЙСТВИЕ КАЖДОЙ ПРОЦЕДУРЫ ПРЕПОДАВАТЕЛЮ!

2. В начало модуля, содержащего программу для кнопки Тестирование, ввести текст программы Part_N, инициализирующей работу ИС:
DefInt I-M
DefSng A-H, O-Z
Public Part As Byte 'Номер темы для тестирования и количество тем
Public Npart As Byte '
Public n1 As Byte ' номера первого и последнего вопросов в теме
Public n_qust_part As Byte
Public nstr As Byte ' номер строки экзаменуемого студента на листе Список студентов
Sub Part_N()
Worksheets("Вопросы").Activate
Npart = Cells(1, 5) ' количество тем в базе вопросов
N_qust = Cells(1, 4) ' считываем количество вопросов в базе
n_qust_part = N_qust / Npart ' число вопросов в данной теме
Select Case Part
Case 1
n1 = 1
Case 2
n1 = n_qust_part + 1
End Select
UserForm2.Show
End Sub
Программа работает с листом Вопросы – база вопросов с вариантами ответов. Содержимое базы уже должно быть скопировано из книги «База_вопросов». Структура базы показана на рис. 3.3.

Рис. 3.3 Фрагмент базы вопросов

В столбце А записаны вопросы с вариантами ответов. В столбце В записаны – правильный ответ (правильный номер). В ячейке D1 количество вопросов в базе. В ячейке Е1 – количество тем в базе, соответствующее количеству переключателей Формы1. (Столбец С – вспомогательный – определяет количество непустых строк в столбце А, по нему рассчитывается количество вопросов в базе.)
ПРОЧИТАЙТЕ ТЕКСТ ПРОГРАММЫ Part_N И РАЗБЕРИТЕСЬ В СМЫСЛЕ ИСПОЛЬЗУЕМЫХ В ПРОГРАММЕ ПЕРЕМЕННЫХ.
Программа Part_N  вызывает следующую Форму2.

[bookmark: _Toc381225131]III. Вторая пользовательская форма – регистрация студента 
для тестирования

Создайте Форму2, показанную на рис. 3.4. В ней используются элементы – Текстовое поле и Кнопка.

Рис. 3.4. Форма для регистрации студента

Текст программ, обрабатывающих кнопки, следующий:

Private Sub CommandButton1_Click()
fam = TextBox1.Text
If fam = "" Then
MsgBox "Введите фамилию еще раз и повторите команду", vbExclamation
TextBox1.SetFocus
Exit Sub
End If
Worksheets("Список студентов").Activate
nst = Cells(1, 7)
If nst = Empty Then ' заполнение шапки таблицы
Cells(1, 1) = "Число студентов": nst = 0
Cells(2, 1) = "ФИО": Cells(2, 2) = "Дата":
For i = 1 To 3: Cells(2, i + 2) = i: Next
Cells(2, 6) = "Оценка"
' чтение количества протестированных студентов
Else
For i = 3 To nst + 2
If UCase(Trim(fam)) = UCase(Trim(Range("a" & i))) Then
ball = Cells(i, 6)
MsgBox "Вы уже сдавали экзамен!" & Chr(13) & "Ваша оценка " & ball _
, vbInformation + vbOKOnly, "Результат поиска"
Exit Sub
End If
Next
End If
' появляется очередной студент, для него готовится строка с фамилией
nst = nst + 1
Cells(1, 7) = nst
nstr = nst + 2  ’Номер текущей строки листа для записи очередного студента
Cells(nstr, 1) = fam
UserForm2.Caption = "Регистрация студента"
UserForm3.Show
End Sub

Private Sub CommandButton2_Click()
UserForm2.Hide
End Sub

ОБРАТИТЕ ВНИМАНИЕ, как в программе для первой кнопки осуществляется проверка содержимого текстового поля: не допускается оставлять его пустым. Прежде чем будет начато тестирование нового студента, его фамилия сравнивается со списком фамилий на листе Список студентов и, если эта фамилия уже встречалась, то КАКОВА РЕАКЦИЯ ПРОГРАММЫ?

[bookmark: _Toc381225132]IV. Третья пользовательская форма – формирование «электронного билета», ввод ответов тестируемого и проверка правильности его ответов

Создайте Форму3, показанную на рис. 3.5. В ней используются элементы – Надпись, Текстовое поле и Кнопка. Для инициализации формы запускается программа, формирующая случайным образом список из трех вопросов выбранной темы. 
[image: ]
Рис. 3.5. Форма3 для генерации «электронного билета»

Private Sub userform_activate()
'Генерация билетов
' чтение данных с листа группы

Randomize Timer
Sheets("Вопросы").Activate

For i = 1 To 3
m1: NVopr(i) = Fix(Rnd * n_qust_part) + n1 ' номер вопроса из соответствующей части
For j = 1 To i - 1
If NVopr(j) = NVopr(i) Then GoTo m1 'исключение повторяющегося вопроса
Next j
Next i
Rem Чтение соответствующего вопроса с листа и ввод его в форму
For i = 1 To 3
k = NVopr(i)
Vop(i) = Range("A" & k)
Otv(i) = Range("B" & k)
Select Case i
Case 1
Label1.Caption = " 1) " + Vop(i)
Case 2
Label2.Caption = " 2) " + Vop(i)
Case 3
Label3.Caption = " 3) " + Vop(i)
End Select
Next i
Rem чтение ответов с формы
Set inpB(1) = TextBox1
Set inpB(2) = TextBox2
Set inpB(3) = TextBox3
For i = 1 To 3: inpB(i) = Empty: Next i
TextBox1.SetFocus
UserForm3.Caption = "Сгенерированный билет"
End Sub

ПРОЧТИТЕ ТЕКСТ ПРОГРАММЫ И РАЗБЕРИТЕСЬ В ЕЁ АЛГОРИТМЕ!

Далее представлены тексты программ, обрабатывающих кнопки формы. Операторы «шапки» программы должны, естественно, стоять в начале модуля.

Dim NVopr(1 To 3) As Byte
Dim Vop(1 To 3) As String
Dim Otv(1 To 3), Otvs(1 To 3), raz(1 To 3) As Byte
Dim inpB(1 To 3) As Object
Private Sub CommandButton1_Click() 'Клавиша верификации ответов и записи их на лист
For i = 1 To 3
'Проверка ввода цифр в поля ответов
[bookmark: OLE_LINK1]If IsNumeric(inpB(i).Text) = False Then
MsgBox "В поле ответа - не цифра", vbCritical
inpB(i).SetFocus
Exit Sub
End If
Next
ball = 2
For i = 1 To 3
Otvs(i) = Int(inpB(i).Text)
difo = Otvs(i) - Otv(i)
If difo = 0 Then raz(i) = 1 Else raz(i) = 0
ball = ball + raz(i)
Next i
Nsh = Ngr + 2
Sheets("Список студентов").Activate
'Запись результатов ответа в активный лист
Cells(nstr, 2) = Date
For i = 1 To 3: Cells(nstr, i + 2) = raz(i): Next
Cells(nstr, 6) = ball
Button = MsgBox("Получен балл = " & ball & " Выйти из Формы?", vbYesNo)
If Button = vbYes Then
UserForm3.Hide
Else: MsgBox "Нажмите другую кнопку!", vbInformation
End If
End Sub
Private Sub CommandButton2_Click()
'если студент отказался отвечать, ему ставится 2
Sheets("Список студентов").Activate
ball = Cells(nstr, 6)
If ball = Empty Then Cells(nstr, 6) = 2
UserForm3.Hide
End Sub

Запустите проект, нажав кнопку Тестирование  на Главной форме. Пройдите тестирование с разными фамилиями по одной теме, затем – по другой теме.

[bookmark: _Toc381225133]V. Расширение списка тем тестирования и соответствующее дополнение базы вопросов 

Просмотрите текст программы и подумайте, какие операторы надо изменить/вставить, если в базу вопросов будет добавлена еще одна тема (состоящая из 5 вопросов). 
Дополните базу следующими вопросами с 4 вариантами ответов (правильный ответ обозначен знаком +)  и модифицируйте программу.  
Совет! Если лень вручную переводить представленные вопросы в формат листа «Вопросы», можете составить независимую программу, которая убирает знаки «+» и «-» из вариантов ответов, ставит вместо них соответствующие номера, создает единую символьную строку вопрос-ответ и записывает его и правильный вариант ответа в соответствующие строки и столбцы листа «Вопросы».

1. Представителем первого поколения ЭВМ был:
-: машина Тьюнинга-Поста
+: ENIAC
-: CRONIC
-: Паскалина
2. Электронная лампа в качестве элемента вычислительного устройства впервые использовалась
-: в первых арифмометрах
-: в персональных компьютерах системы Apple
+: в электронно-вычислительных машинах первого поколения
-: в карманных калькуляторах
-: в вычислительных машинах серии ЕС ЭВМ
3. Первым персональным компьютером был
-: ЕС ЭВМ
+: Apple-I
-: ENIAC
-: EDVAC
4. Первая ЭВМ по схеме фон Неймана была построена в
-: 1961 г.
+: 1946 г.
-: 1954 г.
-: 1834 г.
5. В машине БЭСМ-6 один байт равнялся
-: 4 бита
+: 6 бит
-: 8 бит
-: 10 бит

Поработать с построенным проектом по всем трем темам. Показать результаты работы преподавателю. 

* * *

[bookmark: _Toc381225134]Контрольные вопросы
1. Программа Sub userform_activate(). Объяснить действие оператора 
m1: NVopr(i) = Fix(Rnd * n_qust_part) + n1
2. Программа Sub CommandButton1_Click() Формы 3. Что означает условие IsNumeric(inpB(i).Text) = False в первом операторе IF?
3. Какие методы используются для обращения к разным листам книги в одной программе?

